

Dash and Double-Dash, Shorthand for Variable Lists¹

In SAS, dashes can be used to refer to a list of variables without having to type the name of each variable. With the use of dashes, and double-dashes, SAS provides flexibility in what variables will be specified in the list. This shorthand method can be used within procedures, in keep and drop statements, and in arrays.

The general rule for dash and double dash is:

A single dash is used to specify consecutively numbered variables.

A double dash is used to specify variables based on the order of the variables as they appear in the file, regardless of the name of the variable.

The following program will be used to demonstrate how the symbol “dash” can be used to specify a list of variables. Each line will be explained, and then the results will be shown.

```
title1 'DASH.SAS, A lesson in Dash and Double Dash' ;
data dash ;
input v1 - v4 weight v5 $ gender $ v6 ;
cards ;
1 2 3 4 250 five M 6
;
proc contents varnum ; run ;
title2 'Proc Print, all variables, in order' ; proc print ; run ;
title2 'proc print ; var v1 - v6;' ;          proc print ; var v1 - v6 ; run ;
title2 'proc print ; var v1 -- v6;' ;          proc print ; var v1 -- v6 ; run ;
title2 'proc print ; var v1 -numeric- v6;' ;   proc print ; var v1 -numeric- v6 ; run ;
```

Note the dash in the following line. This will create variables v1 v2 v3 and v4. A dash in an input statement will work only to create sequentially numbered variables.

```
input v1 - v4 weight v5 $ gender $ v6 ;
```

The following line asks SAS to provide information about the data set and to list information about variables in the order of the variable number. (I have abbreviated the output.) `proc contents varnum ; run ;`

DASH.SAS, A lesson in Dash and Double Dash

Data Set Name:	WORK.DASH	Observations:	1
Member Type:	DATA	Variables:	8

-----Variables Ordered by Position-----

#	Variable	Type	Len
1	v1	Num	8
2	v2	Num	8
3	v3	Num	8
4	v4	Num	8
5	weight	Num	8
6	v5	Char	8
7	gender	Char	8
8	v6	Num	8

¹Prepared by Patty Glynn , University of Washington, 03/03/03, revised 3/05/03.

The following proc print will ask SAS to print all variables, in the order that they were created.

```
proc print ; run ;
```

Proc Print, all variables, in order

Obs	v1	v2	v3	v4	weight	v5	gender	v6
1	1	2	3	4	250	five	M	6

The following proc print will ask SAS to print only variables numbered sequentially between v1 and v6.

```
proc print ; var v1 - v6 ; run ;
```

```
proc print ; var v1 - v6;
```

Obs	v1	v2	v3	v4	v5	v6
1	1	2	3	4	five	6

The following asks SAS to print all variables in the file, beginning with v1 and ending with v6, regardless of the name or type of variable. `proc print ; var v1 -- v6 ; run ;`

```
proc print ; var v1 -- v6;
```

Obs	v1	v2	v3	v4	weight	v5	gender	v6
1	1	2	3	4	250	five	M	6

The following asks SAS to print all NUMERIC variables in the file, beginning with v1 and ending with v6, regardless of the name. `proc print ; var v1 -numeric- v6 ; run ;`

```
proc print ; var v1 -numeric- v6;
```

Obs	v1	v2	v3	v4	weight	v6
1	1	2	3	4	250	6

Following is a second part of the program, that shows another context in which dashes can be used.

```
data dash2 ; set dash ;
** variables in an array must be of the same type, so v5 cannot be included ;
ARRAY new n1 - n4 n6 ;
ARRAY old v1 - v4 v6 ;
do over new ;
new = old * 2 ;
end ;
title2 'proc print ; var n1 - n4 n6 ;' ; proc print ; var n1 - n4 n6 ; run ;
title2 'proc print ; var v1 - v4 v6 ;' ; proc print ; var v1 - v4 v6 ;
run ;
```

```
proc print ; var n1 - n4 n6 ;
```

Obs	n1	n2	n3	n4	n6
1	2	4	6	8	12

```
proc print ; var v1 - v4 v6 ;
```

Obs	v1	v2	v3	v4	v6
1	1	2	3	4	6